# Maintainer: Konstantin Podsvirov <konstantin@podsvirov.su>

_realname=pyinstaller
pkgbase=mingw-w64-${_realname}
pkgname=("${MINGW_PACKAGE_PREFIX}-${_realname}")
pkgver=6.13.0
pkgrel=2
pkgdesc='Bundles a Python application and all its dependencies into a single package (mingw-w64)'
arch=('any')
mingw_arch=('mingw64' 'ucrt64' 'clang64' 'clangarm64')
url='https://pyinstaller.org/'
msys2_repository_url='https://github.com/pyinstaller/pyinstaller'
msys2_references=(
  'purl: pkg:pypi/pyinstaller'
)
license=('custom')
depends=("${MINGW_PACKAGE_PREFIX}-binutils"
         "${MINGW_PACKAGE_PREFIX}-pyinstaller-hooks-contrib"
         "${MINGW_PACKAGE_PREFIX}-python-altgraph"
         "${MINGW_PACKAGE_PREFIX}-python-packaging"
         "${MINGW_PACKAGE_PREFIX}-python-pefile"
         "${MINGW_PACKAGE_PREFIX}-python-pywin32-ctypes"
         "${MINGW_PACKAGE_PREFIX}-python-setuptools")
makedepends=("${MINGW_PACKAGE_PREFIX}-python-build"
             "${MINGW_PACKAGE_PREFIX}-python-installer")
options=('!strip')
source=("https://pypi.org/packages/source/${_realname::1}/${_realname}/${_realname}-${pkgver}.tar.gz")
sha256sums=('38911feec2c5e215e5159a7e66fdb12400168bd116143b54a8a7a37f08733456')

if [[ ${CARCH} == aarch64 ]]; then
  source+=("https://files.pythonhosted.org/packages/py3/${_realname::1}/${_realname}/${_realname}-${pkgver}-py3-none-win_arm64.whl")
  sha256sums+=('d9f21d56ca2443aa6a1e255e7ad285c76453893a454105abe1b4d45e92bb9a20')
fi

prepare() {
  cd "${_realname}-${pkgver}"/PyInstaller/bootloader

  # Remove unused bootloaders
  if [[ ${CARCH} == aarch64 ]]; then
    # Add ARM64 bootloader
    cp -r ${srcdir}/Pyinstaller/bootloader/Windows-64bit-arm .

    for bl in $(ls | grep -v Windows-64bit-arm | grep -v images); do rm -rf $bl; done
  else
    for bl in $(ls | grep -v Windows-64bit | grep -v images); do rm -rf $bl; done
  fi
}

build() {
  cp -r "${_realname}-${pkgver}" "python-build-${MSYSTEM}" && cd "python-build-${MSYSTEM}"

  python -m build --wheel --skip-dependency-check --no-isolation
}

package() {
  cd "python-build-${MSYSTEM}"

  MSYS2_ARG_CONV_EXCL="--prefix=" \
    python -m installer --prefix=${MINGW_PREFIX} \
    --destdir="${pkgdir}" dist/*.whl

  install -Dm644 COPYING.txt "${pkgdir}${MINGW_PREFIX}/share/licenses/${_realname}/COPYING.txt"
}
